

ALEXANDRU IOAN CUZA UNIVERSITY OF IAȘI  
FACULTY OF COMPUTER SCIENCE

Program Equivalence: An Interactive  
Relational Separation Logic Prover  
Implemented in Maude

*ANDREI-ALIN CORODESCU*

Session: *July, 2018*

Scientific Coordinator  
*Conf. Dr. Ciobăcă Ștefan*

## **Abstract**

Lucrarea de fata descrie procesul de dezvoltare a unui utilitar interactiv utilizat in analiza formala a programelor, folosindu-se de Logica Separatoare Relationala [1] si Logica Separatoare [2] [3], utilizand Maude [10] drept cadru de dezvoltare. Componenta interactiva se imbina cu o componenta automatizata, prin care parti din demonstratie vor fi realizate automat. Lucrarea este orientata pe rezolvarea problemei prin intermediul functionalitatilor oferite de Maude si va include secvente de cod ce vor ilustra conceptele cheie in modelarea si aplicarea fundamentelor teoretice.

# Contents

<b>1</b>	<b>Contributions</b>	<b>4</b>
<b>2</b>	<b>Description of the problem</b>	<b>4</b>
<b>3</b>	<b>Previous work</b>	<b>4</b>
<b>4</b>	<b>Theoretical Foundations and technologies</b>	<b>5</b>
4.1	Separation Logic . . . . .	5
4.2	Relational Separation Logic . . . . .	5
4.3	Maude . . . . .	6
<b>5</b>	<b>Relational Separation Logic Interactive Prover</b>	<b>6</b>
5.1	Modelling the Separation Logics . . . . .	7
5.1.1	Sort Hierarchy . . . . .	8
5.2	Language Grammar and executable semantics . . . . .	9
5.2.1	Storage Model . . . . .	9
5.2.2	Syntax and Semantics of the language . . . . .	10
5.3	Interaction by Maude LOOP-MODE . . . . .	12
5.4	State of the program . . . . .	12
5.5	Input and output handling . . . . .	13
5.6	Prover execution flow . . . . .	14
5.6.1	User Actions . . . . .	17
5.6.2	System Actions . . . . .	18
5.7	Automated processes . . . . .	19
5.7.1	Variable substitution . . . . .	19
5.7.2	Automatic matching of axioms and previously proven goals . . . . .	19
5.7.3	Automatic demonstration of implications . . . . .	20
5.8	User Interface . . . . .	22

# Introduction

The present paper describes the development of an interactive tool for reasoning how to programs are related, based on studied and previously used theoretical concepts and technologies which facilitate the implementation.

The tool represents an implementation of Hoare Logic - which allows formal reasoning about a program - , along with 2 of its extensions, namely the Separation Logic (named Separation Logic from now on) and Relational Separation Logic [1] (named Relational Logic from now on). The 2 extensions simplify the Hoare Logic proofs, mainly using the "\*" connector, allowing for local reasoning of effects of statements in a program . The tool has been implemented in Maude, a high performance logical framework with powerful metalanguage applications which facilitate the implementations of executable environments for logics.

The tool is built as a CLI which helps [2] [3] [4] [5] [6] [7] [8] [9] with reasoning how two programs are related using Relational Separation Logic specifications. As a consequence of the dependency of Relational Separation Logic on Separation Logic, proofs about single programs using the latter are also supported by the tool. The tool has been developed with extensibility in mind, the main desired extensions being concurrent programs support and automatic proofs.

The rest of the paper is organized as follows:

- **Section 1** will describe the main personal contributions to the realization of this project
- **Section 2** will present the problem this project aims to solve
- **Section 3** will shortly present some other projects related to ours
- **Section 4** will briefly touch upon the theoretical foundations of this project, together with the technologies used to implement them
- **Section 5**, the main section of this paper, will present our approach to solving the problem, and discuss how it can be improved in the future. All the elements of our project will be presented in an individual subsection

# 1 Contributions

Personal contributions to the realization of the project :

- Modelled the Relational Logic and Separation Logic using Maude equational and rewriting logic specifications .
- Developed an interactive tool for reasoning about program behaviour using the aforementioned logics.
- Automation of some tasks which makes the tool more convenient to use .
- Examples of formal proofs done using the tool

# 2 Description of the problem

The problem this project is aiming to solve is related to formal reasoning about the execution of code, mainly focusing on how two programs are related to each other (most often the relation to be proven is equivalence)

Comparing programs or code fragments and studying their equivalence is part of every software engineer's activities when they are testing an alternative implementation for an existing solution, fixing bugs, launching new product versions, etc . Naturally, for every process completed manually there are efforts being made in order to make it more efficient, less error-prone and, in the end, automate the process all together. Once such a task is automated in software engineering, it can be included in the flow of any research or development phase of a product. An example benefiting from a formal proof of program equivalence is compiler optimization, where the optimized code needs to be equivalent to the input one .

This project aims to lay the foundations of a tool which facilitates formal reasoning on the way two programs are related with a focus on extensibility and automation of tasks.

# 3 Previous work

Previous work related to the topic has been done mostly in terms of Separation Logic based tools, with Relational Separation Logic not being treated

as much.

A notable example which also uses the same framework as this project, namely Maude, is the Java+ITP[4] tool, which enables analysis of Java programs using Separation Logic. The implementation relies heavily on Maude's ITP (iterative theorem prover).

## 4 Theoretical Foundations and technologies

This section will briefly introduce the concepts on which the project is based, together with the technologies used to implement it, along with a few references to resources which explain them in depth.

### 4.1 Separation Logic

Separation Logic allows for reasoning about the effects of code on the program state in a formal manner. The main abstraction used in separation logic is the **Hoare Triple**

$$\{P\} C \{Q\}$$

where  $P$  denotes the a condition satisfied by state of the program before the execution of the command  $C$  and  $Q$  denotes another condition satisfied by state of the program after the execution of  $C$ .

Good references for reading on separation logic include :

- [2] - for a good introduction and some interesting usages
- [3] - for a more in depth explanation, including the mathematical fundaments of the logic

### 4.2 Relational Separation Logic

Relational Separation Logic, which is the central concept of our project builds upon the Separation Logic to reason about how to programs are related, by using the concept of a **Hoare Quadruple**

$$\{R\} \begin{array}{c} C_1 \\ C_2 \end{array} \{T\}$$

$R$  represents a **relation** between the two program states holding before the execution of commands  $C_1$  and  $C_2$  respectively while  $T$  represents a **relation**

which holds after the execution of the two commands.

Relational Separation Logic is presented in detail in [1].

### 4.3 Maude

Maude [10] is a framework based on rewriting logic [11] which allows for natural representations of a wide range of applications, including other logics, which made it a perfect fit for our case. Maude allows for short and simple, yet clever and self-explanatory solutions to problems.

Good references for Maude language include:

- Maude Primer [2] which was written for Maude 2.0.1 but still applies to the latest versions of Maude is a good starting point for learning Maude, as it introduces the language in a more informal, friendlier way
- Maude Manual [7] which presents the whole Maude system in depth, along with the mathematical foundations
- All about Maude book [12] which includes everything contained in the manual and presents some of the more relevant tools implemented in Maude .

## 5 Relational Separation Logic Interactive Prover

This section will include details related to the implementation and usage of the Relational Separation Logic prover and highlight a few of Maude's features which made it the a language fit for the purpose of this project.

The information presented in this section assumes the reader's familiarity with Maude, especially the META-LEVEL module and with the underlying theoretical concepts implemented by the prover (Separation Logic and Relational Separation Logic). References for these can be found in the previous section.

This section will include snippets of Maude code, but since Maude is a very expressive and self-explanatory language, I have not included of thorough presentation of every language construct defined for the scope of this project. Ideas presented will be accompanied by some examples and key ideas of the implementation but for a more detailed presentation, please refer to the code itself.

## 5.1 Modelling the Separation Logics

Thanks to Maude's ability to naturally represent logics, the modelling of the Separation Logic and Relational Separation Logic is mostly reduced to translating the logical formulae into appropriate syntax for Maude, making the representational distance (the amount of information lost when representing a theoretical concept into a programming language)[7] almost non existent.

Each proof rule in the Separation Logic and Relational Separation Logic has been translated into a Maude rewrite rule. The prover is centered around the concept of **Goals** which can be one of the three things:

- Hoare Quadruple - corresponding to Relational Separation Logic
- Hoare Triple - corresponding to Separation Logic
- Implication between relations / assertions

The proof rules are interpreted in a bottom-up manner, meaning that applying a rule to a goal matching the conclusion of a rule will generate new goals consisting of the hypothesis elements of the rule : Figure 1 presents how a Separation Logic rule has been translated into Maude syntax and Figure 2 presents the same process for the Relational Separation Logic counterpart of the rule.

Consequence Rule

$$\frac{P \Rightarrow P' \quad \{P'\} C \{Q'\} \quad Q \Rightarrow Q'}{\{P\} C \{Q\}}$$

Is represented in Maude syntax as :

```
1  rl [SeparationLogic-Consequence] : {P} C {Q} => ((P => P1) <>
    ({P1} C {Q1})) <> (Q1 => Q) [nonexec] .
```

Listing 1: Separation Logic Consequence rule

Figure 1: Example of a translation to Maude syntax of a Separation Logic proof rule

The <> operator is used to separate sub-goals generated by the rule.



Consequence Rule

$$\frac{R \Rightarrow R_1 \quad \{R_1\} \mathcal{C}' \{S_1\} \quad S \Rightarrow S_1}{\{R\} \mathcal{C}' \{S\}}$$

Is represented in Maude syntax as :

```

1  rl [Consequence] : { R } C1 — C2 { S } => ((R => R1) <> ({R1}
2    C1 — C2 {S1})) <> (S1 => S) [nonexec] .

```

Listing 2: Relational Separation Logic Consequence rule

Figure 2: Example of a translation to Maude syntax of a Relational Separation Logic proof rule

The `--` operator is used to separate the two snippets of code mentioned in the **Hoare Quadruple**.

All the rules are marked with the attribute `[nonexec]` because we want them to be applied in a controlled manner, at the meta-level. Some of the rules, the examples included, are not even valid executable Maude rules, because they make use of variables before binding them (for example `P1` is used in a rule before it is bound). At the meta-level however we can manually bind values to those variables, as we will exemplify when discussing how we apply these rules to goals.

### 5.1.1 Sort Hierarchy

Much in the same manner as we modeled the language grammar, the Relational Separation Logic [1] and Separation Logic [3] [2] have been modeled by defining sorts and declaring operators using these sorts. The main point of reference for our modeling has been [1], for both logics. Some examples of translations from specification to code are included below.

- **Assertion** is the sort used to handle Separation Logic assertions. It is a super sort of the **BExp** sort defined as part of our language.
- **HoareTriple** is the sort associated with the concept with the same name, in a Separation Logic context.

```

1  op {_}_[_] : Assertion Command Assertion -> HoareTriple .

```

Listing 3: Hoare Triple constructor operator

- **Relation** is the sort associated with Relational Separation Logic relations. As it was the case with **Assertion**, it is a super sort of the **BExp** sort of our language grammar. Because both the **Relation** and **Assertion** are super sorts of the same sort, Maude will display a few warnings which are not affecting functionality, but in the future, creating a separate **BExp** sort for both the **Assertion** and **Relation** sorts would be a better option.
- **HoareQuadruple** is again associated with the concept with the same name, in the Relational Separation Logic context.

```

1  ——— Operator which creates a relation from two assertions
2  op [_;_] : Assertion Assertion -> Relation .
3  ——— Operator which constructs a Hoare Quadruple
4  op {_}_—_{_} : Relation Command Command Relation ->
   HoareQuadruple .

```

Listing 4: Examples of Relational Separation Logic specific constructs

The other operators defined for the Assertion and Relation languages have been in the same manner as the programming language operators, following their presentation in [1]. For a full reference, please refer to the code accompanying this paper.

## 5.2 Language Grammar and executable semantics

The prover supports expressing programs in a simple, imperative language, commonly used throughout papers [3] [2] related to the subject of program verification, including the one describing Relational Separation Logic [1], around which this project is based.

### 5.2.1 Storage Model

A state of in our storage model is defined by a pair consisting of a *Store* and a *Heap*.

Assuming that all the variables usable in programs from the set **Vars** and the set of positive natural numbers is denoted by **PosNats**:

A *Store* is defined as:

$$S : \text{Vars} \rightarrow \text{PosNats}$$

A *Heap* represents a mapping from the **PosNats** to **Integers**

$$H : \text{PosNats} \rightarrow \text{Integers}$$

More informally, the *Store* holds the value of the variables while the *Heap* maps the active memory cells during a program execution to their contents.

### 5.2.2 Syntax and Semantics of the language

---

Integer Expressions	$E ::= x \mid Integer \mid E \text{ plus } E \mid E \text{ times } E \mid E \text{ minus } E$
Boolean Expressions	$B ::= false \mid true \mid B \ \&\& \ B \mid B \    \ B \mid B \ - \> \ B \mid B \ <=> \ B \mid !B \mid E \ \text{ge} \ E \mid E \ \text{le} \ E \mid E \ \text{eqs} \ E$
Commands	$C ::= x := alloc(E) \mid x := [E] \mid [E] := E \mid free(E) \mid x := E \mid C; C \mid \text{if } B \text{ then } C \text{ else } C \mid \text{while } B \text{ do } C \text{ od}$

---

Figure 3: Syntax of the language

The syntax of the language is presented in Figure 3 . It represents an adapted subset of the language presented in the Relational Separation Logic paper [1] .

**Semantics of the language** The semantics of the previously defined language are standard but a few clarifications are necessary to point out how our language constructs interact with the storage model, as well as a few differences from regular programming languages:

- Verbosity of operators - most operators are replaced by their literal names (  $+$  becomes **plus** etc.); we have opted for this approach to avoid complication when implementing the language in Maude, because most of the operators are already defined in Maude for its built-in types, and overloading them could cause conflicts since we are basing our own defined types on Maude's primitives.

$B \ \text{ge} \ B$  translates to the usual  $\geq$  operator

$B \ \text{le} \ B$  translates to the usual  $\leq$  operator

$B \ \text{eqs} \ B$  translates to the usual  $==$  equality comparison operator

$B \ - \> \ B$  denotes implication between boolean expressions

$B \ <=> \ B$  denotes equivalence between boolean expressions

The rest of the operators are mostly similar to their C++ or Java counterparts and their semantics are self explanatory

- Memory allocation / deallocation:

$x := \text{alloc}(E)$  allocates a new cell in the memory, initializes it with the value of  $E$  and stores its address in the variable  $x$

$x := \text{free}(E)$  deallocates the cell at the address equal to the value of  $E$

- Working with variables and memory:

$x := [E]$  reads the contents of the memory cell at address  $E$  and stores the value in the variable  $x$

$[E] := E'$  updates the contents of the memory cell at address  $E$  with the value of the expression  $E'$

$x := E$  updates the value of the variable  $x$  with the value of the expression  $E$  - note that this command does not modify the heap in any way, as it usually happens with regular programming languages when updating the value of a variable

**Maude Modelling of the language** Maude makes the representation of programming languages grammar a straight forward and natural process, by defining sorts corresponding to language elements and their associated operators. The sorts defined in Maude for the grammar are the following :

- **AExp** - abbreviation for Arithmetic Expression - equivalent for Integer expressions in the specification. To capture the first two possible forms of Integer expressions in Figure 3, the AExp sort is a super sort of **Int** of **INT** standard module and **Id**, which is the sort defined for variables.
- **BExp** - abbreviation for Boolean Expressions - equivalent for the Boolean Expressions type in the specification. It is a super sort of **Bool** of standard module **BOOL**.
- **Command** sort maintains the same name as in the specification.

An example of operator translation from specification to Maude code is presented in Listing 5. Note that we have opted to skip the attributes associated with the operators, as they are irrelevant for the purpose of this example.

```

1  op while_do_od : BExp Command -> Command [ ... ].
2  op _:=_ : Id AExp -> Command [ ... ].

```

Listing 5: Example of operator translation to Maude code

**Executable semantics of the language, in Maude** Maude also supports specifying executable semantics for the language using rewriting logic [8]. A proof of concept for a language similar to the one used by the prover is present in the code of the project, but it is out of scope for the presentation of the prover **Need reference to the final file name**. An executable environment for the programming language used in the prover is a valid candidate for future improvements.

### 5.3 Interaction by Maude LOOP-MODE

Maude’s LOOP-MODE module is the entry point for any interactive Maude application, being the only way to store state between commands. In this chapter we will discuss how we handle input and output , the structure of a state and the role every element plays in the execution of the prover.

We will make a distinction between **Loop state**, which specific to Maude’s LOOP-MODE module and the **Prover state** which is specific to our use-case, and holds the necessary data related to the state of the proof we are currently working on. The **Prover state** is encapsulated within **Loop state**.

---

[INPUT, PROVER – STATE, OUPUT]

Figure 4: Loop state

---

### 5.4 State of the program

Following the examples provided together with the Maude Manual [7], our state (Figure 5) is composed multiple sections modified accordingly during program execution .

- **Action** - represents the current action to be executed by the prover, more details about supported actions are included in the following sections.
- **Root Goal** - represents the end-goal of this demonstration

- **Current Goal** - the goal being treated currently - all actions will be applied to this goal
- **Goal Stack** - Stack of goals necessary to be proven in order to prove the root goal. The current goal is the last element popped from the stack and new elements are generated and pushed by the application of proof rules to the current goal.
- **Proven Goals** - List of goals which have been already been proven or axioms. Each goal will be first checked to see if it matches any goal in this list to prevent repeating the same proofs.
- **Staging output** - contains the output generated by the current action, before being passed to the Loop state. This staging mechanism enables each action (user or system) to generate output independent of the other actions.

---

< Action, RootGoal, CurrentGoal, GoalStack, ProvenGoals, StagingOutput >

---

Figure 5: Prover State

---

## 5.5 Input and output handling

To understand how our prover handles input and output, it is important to note that the flows of the LOOP-MODE module is as follows:

1. User input is placed in the **INPUT** member of the loop state
2. Maude engine applies all the rewritings possible, resulting in a new loop state.

This is the step where all the logic of the prover goes into, in the form of rewriting rules which modify the prover state.

3. The member **OUTPUT** of this new loop state is taken and displayed at the console.

There are two main rules used for handling input and output :

1. **in** rule, which is responsible for parsing the user input and modifying the action in the prover state accordingly

```

1  crl [in] :
2  — [ INPUT, <PROVER_STATE>, OUTPUT]
3  [QIL, < noRule ; RG ; G ; GS ; GL ; nil > , QIL']
4  =>
5  if T:ResultPair? :: ResultPair
6  — If parsing succeeded, place the parsed action in
7  — the prover state
8  then [nil, < downTerm(getTerm(T:ResultPair?), noAction) ; RG
        ; G ; GS ; GL ; nil >, QIL']
9  — If the parsing failed, display an error
10 else [nil, < noRule ; RG ; G ; GS ; GL ; nil >, 'ERROR QIL]
11 fi
12 — Try to parse the input as an term of sort Action
13 if QIL /= nil /\ T:ResultPair? := metaParse(upModule('
    PROVER-GRAMMAR, false), QIL, 'Action) .

```

Listing 6: in rule

2. out rule, which takes the staging output from the prover state and appends it to the currently existing output in the loop state

```

1  crl [out] :
2  [QIL, < A ; RG ; G ; GS ; GL ; QIL' >, QIL'' ]
3  => [QIL, < A ; RG ; G ; GS ; GL ; nil >, QIL'' QIL']
4  if QIL' /= nil .

```

Listing 7: out rule

## 5.6 Prover execution flow

In this section we will describe the flow of a demonstration made using our prover. The most important parts of the prover's execution are controlled through the first member of a prover state, **Action**, which specifies the next step in the processing of the current goal. There are two types of actions, **System Actions**, colored in the figure in purple and **User Actions**, colored in light-yellow. The diagram presenting how the prover works is presented in Figure 6

1. The user issues the command `loop init` . in the Maude console. This will initialize the loop state.
2. Now the user will have to specify the goal it wants to prove, by issuing the command

(prove G)

, where  $G$  is the representation of the goal using the syntax defined for our logics. The command will populate the required fields of the prover state (**RootGoal**, **GoalStack**) with  $G$ .

This command will also populate the **ProvenGoals** list with axioms.

Please note that the command is enclosed within parentheses, this being a requirement of Maude LOOP-MODE [7]. All the prover specific commands need to be enclosed in parenthesis. The previous command, `loop init .` was not enclosed because it is a Maude specific command, not something we want to pass as input to our prover.

3. The next phase involves applying the logic proof rules to the goals in the **GoalStack** to generate sub-goals and prove them. Once a goal is popped from the **GoalStack**, it can be replaced by sub-goals generated by proof rules or, matched with an proven goal / axiom or be admitted manually. This phase lasts until there are no more goals left in the **GoalStack**, meaning that the **RootGoal** has been proven.

**Goal Graph** One of the improvements which could make the user experience much better for our prover would be the capability to backtrack in the proof and try another approach for the same goal. The initial idea for this prover was to have a **GoalGraph** instead of **GoalStack** such that when the user applies a rule to a goal, the sub goals are represented as nodes in the same graph, making it possible to trace back the progress to intermediary states. This approach would also make the **ProvenGoals** a lot more useful by increasing the chance of running into an already proven goal and the ability to mark complex goals as proven. At the moment, the only goals which can be used from the **ProvenGoals** are those admitted manually by the user, since the other proven goals already matched axioms.






Figure 6: Prover Execution Diagram

### 5.6.1 User Actions

User actions are the main entry point for user interaction and they describe how the user intends to process the current goal.

- Rule application action - indicates the application of a specific rule to the current goal, the action can be parametrized or not, depending on the rule. These actions are named following the pattern :

$$[\mathbf{SL}] + \mathbf{apply} + \mathit{RuleName}$$

With **SL** being added only when applying Separation Logic rules, to distinguish between the Relational Separation Logic rules with the same name.

```

1  crl [applyConsequence] : < applyConsequence R1 — S1 ; RG
   ; G ; GS ; GL ; nil > =>
2    if SubGoals? :: Result4Tuple then
3      if downTerm(getTerm(SubGoals?), emptyGL) :: GoalList
   then
4        < noAction ; RG ; noGoal ; downTerm(getTerm(SubGoals?),
   emptyGL) <> GS ; GL ; 'applied ' 'consequence ' 'rule
   '\n >
5        else
6        < noAction ; RG ; G ; GS ; GL ; 'did ' 'not ' 'apply ' '
   consequence ' 'rule '\n >
7        fi
8        else
9        < noAction ; RG ; G ; GS ; GL ; 'error ' 'consequence '
   'rule '\n >
10       fi
11  if SubGoals? := metaXapply(upModule('PROVER-INTERFACE,
   false), upTerm(G), 'Consequence, 'R1:Relation <- upTerm(
   R1) ; 'S1:Relation <- upTerm(S1) , 0, unbounded, 0).
12

```

Listing 8: Rule Application Action

The example given in Listing 8 illustrates all the core concepts involved in the application of proof rules to goals:

- The rewrite rule is applied to a **Prover state** which has the Action field equal to **applyConsequence R1 - S1**, where R1 and S1 are parameters to be passed to the proof rule
- The Consequence rule is applied at a meta level, using the **metaXapply** operator [line 11] from the **META-LEVEL** module. To this operator

we also pass the values of the parameters (in their meta representation) with which to bind variables appearing in the **Consequence** rule.

- The result is then checked for errors to see if there were some errors when applying the **metaXapply** operator [line 8] or if the term returned is not actually of the required type [line 3].
- If the result corresponds to what the prover was expecting, we modify the state by adding the newly generated goals to the **Goal-Stack**, by moving down the "meta"-hierarchy - parsing meta-representation of terms.
- The last thing to note is an example of how the rules generate output independent of other applications. In the last field of the **Prover State**, as **StagingOutput**, there will be a list of quoted identifiers which will eventually get moved to the loop state and then outputted to the user.

All the other actions which control the application of proof rules are written in a similar manner.

- **ok** action - admits the current goal
- **showGoal** action - prints the current goal to the console - it does not modify the state of the prover
- **auto x** action - instructs the prover to attempt to automatically prove an implication, search for a solution up to a depth of **x** rewrites

### 5.6.2 System Actions

System actions were implemented to mimic the behaviour of a "chain of responsibility" design pattern from object-oriented programming. The system actions are all applied in order before asking for user input. After an system action completes execution, it will decide the next action to be placed in the prover state. System actions are the main component through which our prover can be enriched with other features in the future, because integrating a new system action is simple. At the moment, the system actions are:

- **check** - marks the current goal to be checked against previously proven goals or axioms
- **implication** - marks the current goal to be run through the implication prover - in case the goal is an implication, try to prove it automatically

- **stop** - when there are no more sub-goals to be proven, the prover should stop and indicate the success of proving the root goal

## 5.7 Automated processes

This section will describe the functional parts of the prover which have been automated to make it more user-friendly and even prove sub-goals of the main demonstration by itself. Since the automation of the prover is part of the future goals of this project, we will also include a few notes on how we might approach problems related to this goal, even though they have not been implemented yet.

### 5.7.1 Variable substitution

In both sections describing automated processes we will interpret the equality operator between variables and replace, textually, one variable with the other in **Assertions**, **Relations** and **Commands**. The definition of the operator for **Assertions** is presented in Listing 9. The operator is defined recursively, with the base case being treated using the **owise** equational attribute. A similar approach is used for all the definitions of this operator.

```

1 op replace : Assertion Id Id -> Assertion .
2 eq replace(P * Q, X, Y) = replace(P, X, Y) * replace(Q, X, Y) .
3 eq replace(P /\ Q, X, Y) = replace(P, X, Y) /\ replace(Q, X, Y) .
4 eq replace(P \/ Q, X, Y) = replace(P, X, Y) \/ replace(Q, X, Y) .
5 eq replace(emp, X, Y) = emp .
6 eq replace(E |-> E2, X, Y) = replace(E, X, Y) |-> replace(E2, X,
  Y) .
7 eq replace(P, X, Y) = P [owise] .

```

Listing 9: Replace operator definition for assertions

### 5.7.2 Automatic matching of axioms and previously proven goals

The prover is capable of automatically recognizing goals matching an axiom or a previously proven goals, for a better user experience. It achieves this goal by using **metaXmatch** operator included in the standard **META-LEVEL** module, which determines if two terms match.

- In terms of previously proven goals, the process is straight-forward, as it will compare terms in a literal manner (since they do not contain any variable)

- Axioms are modeled as terms with variables in them, allowing them to be matched with any term for which all variables in the axiom can be bound. For example, the axiom

$$\{E \mapsto -\}[E] := F\{E \mapsto F\}$$

will be represented as a term of sort `HoareTriple`, but with free variables:

```
1 {E: AExp |-> - } [E: AExp] := F: AExp {E: AExp |-> F: AExp}
```

However, Maude does not allow for variables to appear for the first time on the right hand side of a rewrite rule which needs to be used without meta-level computations (problem present when modeling the Separation Logics as well). We had to find a work around for this problem and the solution we came up with was to write the axiom in its meta-representation, and then use the `downTerm` operator to bring it down in the meta-hierarchy (Listing 10)

```
1 rl [proveGoal] : < prove(HQ) ; RG ; G ; GS ; GL ; nil > =>
2 < noAction ; HQ ; noGoal ; HQ <> emptyGL ;
3 downTerm('{'_'}{'_' }['_']|>['_'] 'E: AExp, '_' : AExp] , '['_'] := ['E:
  AExp, 'F: AExp] , '['_']|>['_'] 'E: AExp, 'F: AExp]] , emptyGL)
4 — ... [other axioms omitted for clarity]
5 ; 'begin '\n > .
```

Listing 10: Initialization of Prover state

- For matching we also interpret the equality between variables using the `replace` operator introduced previously to reduce the `Goal` to a simpler form, which might match to an axiom.

```
1 rl {P * (emp /\ X eqs Y)} X1 := E {Q} =>
2 if X1 /= X then
3 {replace(P, X, Y)} replace(X1 := E, X, Y) {replace(Q, X, Y)}
4 else
5 {replace(P, X, Y)} X1 := replace(E, X, Y) {Q}
6 fi .
7
```

Listing 11: Rule interpreting variable equality

### 5.7.3 Automatic demonstration of implications

Automatic demonstration of implications is one of the most challenging aspects of this paper and this section will present our incipient approach to this problem.

We have modeled implication proof rules as rewrite rules in Maude (example in Listing 12), in order to make use of its `metaSearchPath` operator, which will search for a series of rewrites which transform a term into another. We used this functionality to try and solve implications

$$R \Rightarrow S$$

by searching for a series of rewrites from `R => S` to `true`.

```

1  rl R:Relation * ( S:Relation \/ T:Relation ) => (R:Relation * S:
   Relation) \/ (R:Relation * T:Relation) .
2  ——— Example for the usage of the replace operator
3  rl R * (Emp /\ X eqs Y) => replace(R, X , Y) .
4  rl (R:Relation => R:Relation) => true .

```

Listing 12: Example of implication rewrite rules

```

1  op tryImplicationProving : Goal Int -> Trace? .
2  eq tryImplicationProving(RR:Relation => TT:Relation , depth:Int) =
   metaSearchPath(upModule('IMPLICATION-PROVER, false), upTerm(
   RR:Relation => TT:Relation), upTerm(true), nil, '* , depth:Int
   , 0) .

```

Listing 13: Equation using the `metaSearchPath` operator

In its current form, it cannot offer any guarantee, except for the certainty that if the implication has been proven automatically, the proof is correct in respect to the rewrite rules specified in the **implication-prover.mau** file. The approach however has an exponential running time since it searches exhaustively throughout the state space for a solution. The second parameter is there to control the depth up to which the search goes. The prover will try automatically to search a solution within a depth of 3 rewrites, but the user can manually specify searching for a solution using the `auto x` command, where `x` is the depth up to which Maude should try to search the solution (our experiments show that searches up to a depth around 12 should finish use a reasonable amount of resources and finish in under 10 minutes - this observation of course depends on the implication and the number of possible rewrites starting from it). The found path is presented to the user if the implication has been proven.

**Inductive predicates** - the prover currently supports defining inductive predicates only by modifying the underlying Maude code. We have defined a single predicate, used in the Relational Separation Logic paper [1] :

```

1  op List : Id -> Relation .

```

Listing 14: Predicate syntax definition

In terms of the semantics of the `List` predicate, we have added rules to the **implication-prover.maude** file, to be considered when rewriting implications.

```
1 r1 [ (X:Id |-> E:AExp) * ((X:Id plus 1) |-> Y:Id) ; (X:Id |-> E:
    AExp) * ((X:Id plus 1) |-> Y:Id) ] * List(Y:Id) => List(X) .
```

Listing 15: Example of rule interpreting the `List` predicate during implication proving

**Cyclist** [9] [13] is an automatic entailment prover for Separation logic which we planned to integrate with our prover for separation logic implications and eventually extend to Relational Separation Logic, since it provides a generic framework. Since Maude does not easily support external commands, the idea would be to modify Maude’s C++ code and add a hook for the `[special]` operator attribute which would invoke that hook, which will call the external command for Cyclist. We have however focused on other aspects of our prover, and left this feature for the future.

## 5.8 User Interface

Since our project aims to simplify the demonstrations of Separation Logic / Relational Separation Logic formulae, using an interactive prover, the user interface plays a major role in the success of the project - the easier it is to use and understand, the more chances of success it has. This idea has been presented in [2], when discussing key elements of tools related to program verification.

Maude provides a simple, yet powerful system for controlling the output formatting of the CLI through which we aim to make the demonstrations more readable and user-friendly.

The Maude features we used for this prover’s user interface are :

- The **format** attribute of operators, which allows the programmer to control the style of the output when a term which makes use of the respective operator is printed to the console. It allows specifying modifiers for each possible white space in an operator, modifiers which will remain active until reset or until overridden. More information on the **format** attribute can be found in the Maude Manual [7].

A sample output of the prover is presented in Figure 7, showcasing the manipulation of white spaces (newlines, indentation and spaces),

usage of colours to distinguish between different elements - Relations are coloured in green, while the analysed code has it's keyword highlighted using cyan. The separator between the two pieces of code is also coloured in red.

```
{ Same * List(c) * ( Emp /\ y eqs yp ) * ( Emp /\ c eqs cp ) * [y l-> x0 ; yp l-> x0] }
pass ;

while !(c eqs null) do
  x := [y];
  [c] := (0 minus x);
  c := [c plus 1]
od
--
xp := [yp];

while !(cp eqs null) do
  [cp] := (0 minus xp);
  cp := [cp plus 1]
od
{ Same * ( Emp /\ y eqs yp ) * ( Emp /\ c eqs cp ) }
```

Figure 7: Formatted output of the prover

- The `metaPrettyPrint` operator in the `META-LEVEL` module, which returns a term as a list of quoted identifiers to be printed, formatted respecting the `format` attributes defined for the operators used in the term. An example is presented in Listing 16, which presents the rule used to take the top element of the **GoalStack** for processing, and writes the formatted output of the goal to the **StagingOutput** field.

```
1 crl [takeFirstGoalFromList] : < noAction ; RG ; noGoal ; G <> GS
  ; GL ; nil >
2 =>
3 < check ; RG ; G ; GS ; GL ; 'current ' 'goal ' ' 'is '\n
  metaPrettyPrint(upModule( 'PROVER-INTERFACE, false ), upTerm(G)
  ) '\n >
4 if G /= noGoal .
5
```

Listing 16: Rewrite rule making use of `metaPrettyPrint`

All throughout the prover calls to `metaPrettyPrint` have been made in order to inform the user of what the prover is doing and how.

For a more advanced user interface or to expose the prover's functionalities through a programming interface, a wrapper written in more feature-rich



language through which to control the Maude environment is probably the best choice. The concept is exemplified by the Maude plug-in for Eclipse, [14] which enables writing of Maude code inside the Eclipse IDE, and the plug-in communicates with the underlying Maude environment. The project also includes a plug-in which allows any Java program to communicate with Maude.

## Conclusions

The development of the Relational Separation Logic prover helped us get a deeper understating of it, along with an introduction to other theoretical concepts used in program analysis and programming language semantics, such as Separation Logic and Rewriting Logic.

While the two Separation Logics used throughout the paper are powerful tools for analyzing the behavior of program and are relatively easy to understand, it takes time to become familiar with them and use them at their full potential.

The logics are better suited for studying small snippets of code and not entire programs, because the demonstration can get cumbersome and end up spending a lot of time on focusing on irrelevant parts of the program.

Maude has been interesting to work because it required an entirely different mindset than the other programming languages I was familiar with. Once getting grasping the concepts around Maude which was built, it was easy to come up with simple, clever solutions to the problems I was facing. As it is described on its website [10], Maude is a powerful logical framework which allows modeling of other logics in it; it proved to be exactly that, enabling us to represent the logics in a natural manner, which made the implementation much less error-prone and allowed us to focus on the relevant part of this project, namely how to apply the logics instead of focusing on how to properly implement them.

While being a powerful and interesting language to work with, being a fairly specialized and not very widespread, guidance and help are limited to the manual [6] and [7]. This obstacle resulted in significant time spent on searching for solutions to otherwise trivial problems, especially when starting using the language, because the source of the problem is usually unclear to a beginner.

Overall, this whole project has been an interesting introduction to both theoretical concepts regarding program analysis and new, different technologies. The mix of the two resulted in the built prover, which could be improved in the future with the features discussed throughout the chapters and others to easily accommodate real-world usage.

## References

- [1] Hongseok Yang. Relational separation logic. *Theor. Comput. Sci.*, 375(1-3):308–334, April 2007.
- [2] Peter W. O’Hearn. A primer on separation logic (and automatic program verification and analysis). In *Software Safety and Security*, 2012.
- [3] J. C. Reynolds. Separation logic: a logic for shared mutable data structures. In *Proceedings 17th Annual IEEE Symposium on Logic in Computer Science*, pages 55–74, 2002.
- [4] Ralf Sasse and José Meseguer. Java+itp: A verification tool based on hoare logic and algebraic semantics. *Electronic Notes in Theoretical Computer Science*, 176(4):29 – 46, 2007. Proceedings of the 6th International Workshop on Rewriting Logic and its Applications (WRLA 2006).
- [5] Traian Florin Șerbănuță, Grigore Roșu, and José Meseguer. A rewriting logic approach to operational semantics. *Information and Computation*, 207(2):305 – 340, 2009. Special issue on Structural Operational Semantics (SOS).
- [6] Theodore McCombs. Maude 2.0 primer, August 2003.
- [7] Manuel Clavel, Francisco Duran, Steven Eker, Santiago Escobar, Narciso Patrick Lincoln, Marti-Oliet and Jose Meseguer, and Carolyn Talcott. *Maude Manual (Version 2.7.1)*, July 2016.
- [8] José Meseguer. Conditional rewriting logic as a unified model of concurrency. *Theoretical Computer Science*, 96(1):73 – 155, 1992.
- [9] James Brotherston, Nikos Gorogiannis, and Rasmus L Petersen. A generic cyclic theorem prover. In *Asian Symposium on Programming Languages and Systems*, pages 350–367. Springer, 2012.

- [10] Maude site. [http://maude.cs.illinois.edu/w/index.php?title=The\\_Maude\\_System](http://maude.cs.illinois.edu/w/index.php?title=The_Maude_System), (accessed 20.06.2018).
- [11] José Meseguer. Twenty years of rewriting logic. *The Journal of Logic and Algebraic Programming*, 81(7):721 – 781, 2012. Rewriting Logic and its Applications.
- [12] Manuel Clavel, Francisco Durán, Steven Eker, Patrick Lincoln, Narciso Martí-Oliet, José Meseguer, and Carolyn Talcott. *All About Maude - a High-performance Logical Framework: How to Specify, Program and Verify Systems in Rewriting Logic*. Springer-Verlag, Berlin, Heidelberg, 2007.
- [13] Cyclist prover. <http://www.cyclist-prover.org/>, (accessed 20.06.2018).
- [14] Maude development tools. <http://mdt.sourceforge.net/>, (accessed 20.06.2018).